
第5章 機械工学編

　機械工学を学ぶ皆さんにとって，学んだ原理や定理，現象を目で見て確

かめたいと思ったことは多いのではないでしょうか．本章では十進BASIC

を用いて，機械工学で学んだことを自宅のパソコンで簡単に調べる練習を

したいと思います．ここでは次の 4テーマを取り上げます．機械工学の計

算をする（調べてみる）基本パターンは同じですので，これらを応用して

他の分野にも取り組んでみて下さい．

1. ピストンクランク機構（機構学・工業力学）

2. はり（材料力学）

3. ガスサイクル（熱力学）

4. 流れの数値シミュレーション（流体力学）

以下ではできるだけ難しい計算法を使わず，また工学式が見通しやすい

プログラムにしてあります．ただし，十進 BASICの基本は学んで頂いて

いるものとします．

5.1 ピストンクランク機構

機械の動きを知ることは，設計製作をする上でとても大切なことです．こ

こでは，簡単なリンク機構の一つであるスライダクランク機構（本節ではピ

ストンクランク機構と呼びます）の動きを調べてみましょう．十進BASIC

を使って，変位・速度・加速度線図を描画してみます．

5.1.1 変位・速度・加速度線図

図 5.1に示すようなピストンクランク機構におけるピストン変位 x[m]・

速度 v[m/s]・加速度 a[m/s2]は，それぞれ次の式により近似的に求められ

ます [1],[2]．

x = r (1 − cos θ) + r

4λ
(1 − cos 2θ) (5.1)

v = ωr

µ
sin θ +

sin 2θ

2λ

¶
(5.2)

a = ω2r

µ
cos θ +

cos 2θ

λ

¶
(5.3)

(5.4)

ここで，r[m]はクランク長さ，θ[rad]は上死点からの回転角，ω[rad/s]は

クランクの角速度を表します．また，λは連かん比と呼ばれ，クランクと

連接棒長さの比を表します．多くの場合には 3～5程度の値をとります．

λ =
l

r

また，ピストンの平均速度 vmean[m/s]は次式で求められます．

vmean =
2SN

60
=
rN

15
(5.5)

S[m]はストローク（=2a），N [rpm]はクランク（エンジン）回転数を表し

ます．

これらの式を使って，変位・速度・加速度線図を描画し，平均速度とコ

ンロッドに作用する最大慣性力を求めるプログラムを次に示します．

100 !---ピストンクランク機構
110 option angle degrees

120 !-変位表示用座標軸

図 5.1: ピストンクランク機構 図 5.2: ピストン速度 図 5.3: ピストン加速度

130 !set window -90, 720, -0.005, 0.1

140 !draw grid(60, 0.01)

150 !-速度表示用座標軸
160 set window -90, 720, -40.0, 40.0

170 draw grid(60, 10.0)

180 !-加速度表示用座標軸
190 !set window -90, 720, -20000.0, 35000.0

200 !draw grid(60, 10000.0)

210

220 declare numeric s, l, r, lambda, n, m, omega, theta, ma_max

230

240 !-関数定義（上死点からのピストン降下距離・速度・加速度）
250 def x(theta)=r*(1.0-cos(theta))+r/4.0/lambda*(1.0-cos(2.0*theta))

260 def u(theta)=omega*r*(sin(theta)+sin(2.0*theta)/2.0/lambda)

270 def a(theta)=omega^2*r*(cos(theta)+cos(2.0*theta)/lambda)

280

290 !-データ入力
300 input prompt "ストローク [mm]を入力して下さい":s

310 input prompt "コンロッド中心間距離 [mm]を入力して下さい":l

320 input prompt "ピストン質量 [kg]を入力して下さい":m

330

340 let r=s/2.0/1000.0

350 let lambda=l/1000.0/r

360

370 !-グラフの描画
380 for n=1000 to 8000 step 1000

390 let ma_max=0.0

400 let omega=2.0*pi*n/60.0

410

420 for theta=0.0 to 720.0 step 10.0

430 ! plot lines: theta, x(theta);

440 plot lines: theta, u(theta);

450 ! plot lines: theta, a(theta);

460 if abs(m*a(theta))>ma_max then let ma_max=abs(m*a(theta))

470 next theta

480 plot lines

490

500 print "平均ピストン速度は", n*s/1000.0/30.0, "[m/s]"

510 print "最大慣性力は", ma_max/1000.0, "[kN]"

520 next n

530

540 end

表 5.1: 変数リスト
s ストローク 入力値
l コンロッド中心間距離 入力値
r クランク長さ 入力値

lambda 連かん比
n エンジン回転数
m ピストン部質量 入力値

omega 角速度
ma max 加速度の最大値

プログラムの説明 120行から 200 行では座標軸の設定を行っています．

表示したいグラフによって適切なものを選択，他はコメントアウトして下

さい．

220行は変数の宣言です．表 5.1に変数一覧を示します．240行から 270

行は関数の定義です． 5.1式から 5.3式までの計算式を定義しています．

290行から 320行まではデータ入力部，340，350行でクランク半径と連か

ん比を求めています．

380行から 520行はエンジン回転数に関するループです．各回転数にお

いて，420から 480行をメインループとし，各線図の描画，最大加速度の算

出を行っています．ここでも，描画させたい関数を選んで他のものをコメ

ントアウトして下さい．また，最大慣性力や平均速度の計算，表示も行っ

ています．

例題 ピストン（ピストン，ピストンピン，ピストンリング，小端部キャッ

プ等含む）質量が 0.41[kg]，ストロークが 75[mm]，コンロッド中心間距離

が 130.5[mm]，最大出力を 6500[rpm]で発生するエンジンがあります．この

エンジンのピストンクランク機構について，各線図を描画してみましょう．

プログラムに入力して速度・加速度線図を描画した結果を図 5.2及び 5.3

に示します．計算式を見直してみると，三角関数部に一次と二次の成分が

あることがわかります．計算結果にもその様子が現れています．この影響

は，連かん比によって変わってきます．連かん比が変わるように入力値を

変えて，その様子を確かめてみて下さい．

また，ピストン平均速度はこの場合，8000[rpm]まで計算すると 20[m/s]

程度まで変化するようです．最大出力のときはどうですか？大体16～17[m/s]

程度の平均速度を持つようですね．

課題 1 図書館やインターネットを使えば（分解して計測してみても良い

ですね），身近なエンジンについて必要なデータを入手することができま

す（例えば [3]）．好みのエンジンについて計算，各線図を描画させてみま

しょう．また，いろんなエンジンのピストン平均速度について検討してみ

ましょう．何か基準のようなものが見つかりませんか？

課題2 最大速度を持つクランク角度は，加速度がゼロになる位置として 5.3

式から解析的に求められます．一方，これらの式は近似式であり，正確な

式は文献 [1],[2]に示されています．これらの場合について，速度最大とな

る角度を求める方法を考えてみてください．

5.2 はり

構造物や機械製品を構成する基本要素の一つに，はりがあります．片持

ちはりや両端支持はりなどは，構造を支える重要な役割を担っています．こ

こでははりについて理解を深めるために，両端支持はりを対象とした計算

を行ってみましょう．

5.2.1 断面二次モーメントと断面係数

はりの設計には，作用する最大曲げ応力を知ることが必要です．最大曲

げ応力 σmax[Pa]は，最大曲げモーメントMmax[N ·m]と断面係数Z [m3]か
ら次式で計算できます [4],[5]．

σmax =
Mmax

Z
(5.6)

したがって，最大曲げ応力を計算するためには，曲げ応力の分布と断面係

数を求めなくてはなりません．また，断面係数は次式で求められます．

Z =
I

ymax
(5.7)

ここで，I [m4]は図形の断面二次モーメント，ymax[m]は最大の端面距離で

す．そこで，まず任意形状の断面二次モーメント及び断面係数を計算する

プログラムを作成してみます．

長方形断面の重心軸回りに関する断面二次モーメントは，次式により算

出できます．

I =
bh3

12
(5.8)

b[m]は長方形の幅，h[m]は高さを表します．

一方，図 5.4に示すような，重心軸から a[m]だけ離れた場所にある長方

形の断面二次モーメントは，次の平行軸の定理により求められます．

I =
bh3

12
+ a2S (5.9)

ここで，S は対象図形の面積です．

S = b × h

(5.9)及び (5.7)式を使って，長方形によって構成される任意形状の断面二

次モーメント及び断面係数を求めるプログラムを次に示します．与えられ

た形状を構成する各長方形の断面二次モーメントの合計が，総合の断面二

次モーメントになります．ただし，簡単のために形状は重心軸に対して対

象であるとします．

100 !---断面二次モーメントと断面係数
110 set window -100, 100, -100, 100

120 draw grid(10.0, 10.0)

130

140 declare external picture Rectangle

150 declare numeric i, n, b, h, ax, ay, ymax, j, s

160

170 input prompt "図形分割数 [-]を入力して下さい":n

180 let ymax=0.0

190 let j=0.0

200 let s=0.0

210

220 for i=1 to n

230 input prompt "新しい分割領域の幅、高さ、水平、垂直距離を入
力して下さい [mm]":b, h, ax, ay

240 if b>0.0 then

250 set area color 1

260 draw Rectangle with SCALE(b/2.0, h/2.0)*SHIFT(ax, ay)

270 else

280 set area color 2

図 5.4: 平行軸の定理 図 5.5: I型断面はり

290 draw Rectangle with SCALE(abs(b)/2.0, h/2.0)*SHIFT(ax, ay)

300 end if

310

320 if (abs(ay)+h/2.0)>ymax then let ymax=abs(ay)+h/2.0

330

340 let s=s+b*h

350 let j=j+b*h^3/12.0+ay^2*b*h

360 print "現在の総面積は", s, "[mm^2]"

370 next i

380

390 print "断面二次モーメントは", j, "[mm^4]"

400 print "断面係数は", j/ymax, "[mm^3]"

410

420 end

430

440 external picture Rectangle

450 plot area: -1.0, -1.0; 1.0, -1.0; 1.0, 1.0; -1.0, 1.0

460 end picture

プログラムの説明 140行は形状を描画する外部関数の宣言です．描画関

数 Rectangleは 440-460行に定義してあります．220行は変数の宣言です．

表 5.2に変数一覧を示します．170行では与えられた形状をいくつの長方形

に分割して計算するかを入力しています．180-200行は変数の初期化です．

表 5.2: 変数リスト
i ループ変数
n 長方形への分割数 入力値
b 長方形の幅 入力値
h 長方形の高さ 入力値

ax 長方形の x座標 描画にのみ使用
ay 長方形の y座標
ymax 最大の端面距離
j 断面二次モーメント
s 面積

220-470行が分割された各長方形に対する入力と計算部です．長方形の

幅と高さを入力し，画面への描画（240-300行），最大の端面距離の評価

（320行），断面二次モーメントと面積の計算（340-360行）を行っていま

す．ただし，穴の場合には幅をマイナスとして入力しています（これでう

まく計算できることを自分で確認してみて下さい）．断面係数は，求まっ

た総合の断面二次モーメントと端面距離から，400行で計算，出力をして

います．

例題 図 5.5 に示すような断面形状のはりについて入力，計算を行って

みましょう．ここでは図形を三分割して考えます．一番目は上端の部分

（b=50mm，h=10mm，ax=0mm，ay=20mm），二番目は中心部（b=10mm，

h=30mm，ax=0mm，ay=0mm），三番目は下端部（b=50mm，h=10mm，

ax=0mm，ay=-20mm）です．計算結果を次に示します．

図形分割数 [-]を入力して下さい 3

新しい分割領域の幅，高さ，水平，垂直距離を

入力して下さい [mm] 50, 10, 0, 20

現在の総面積は 500 [mm^2]

新しい分割領域の幅，高さ，水平，垂直距離を

入力して下さい [mm] 10, 30, 0, 0

現在の総面積は 800 [mm^2]

新しい分割領域の幅，高さ，水平，垂直距離を

入力して下さい [mm] 50, 10, 0, -20

現在の総面積は 1300 [mm^2]

断面二次モーメントは 430833.333333334 [mm^4]

断面係数は 17233.3333333334 [mm^3]

課題1 同じ総面積で最大の断面係数を持つ形状を調査してみましょう．加

工のしやすさなども考えてみて下さい．

課題 2 円も細かく分割すれば長方形の集まりとして考えられます．プロ

グラムを使って，円の断面二次モーメント・断面係数を近似計算してみて

下さい．求まった値を円の公式による結果と比較してみましょう．

課題3 例題プログラムは，ひとまず対照的な図形について考えました．次

は図形の重心も不明として算出し，求まった重心回りの断面二次モーメン

ト，断面係数を計算するプログラムを作成してみて下さい．

5.2.2 せん断力・曲げモーメント・たわみ線図

はりの最大曲げ応力を求めるには， 5.6式で示したように，断面係数と

最大曲げ応力が必要でした．前節では断面係数を求めるプログラムを作成

しました．ここでは，はりのせん断力線図（SFD）及び曲げモーメント線

図（BMD）を描画，最大曲げモーメントを求め，曲げ応力を計算してみま

しょう．また，たわみの基礎式からたわみ線図を描画してみましょう．

図 5.6に示すように，n個の集中荷重が作用する両端支持はりについて

考えます．i番目の荷重 Pi[N]によるせん断力 Si[N]及び曲げモーメント

Mi[N ·m]は，それぞれ次の式で求められます [4],[5]．

Si(x) =

(
RA,i (0 ≤ x ≤ xi)
−RB,i (xi ≤ x ≤ l)

(5.10)

Mi(x) =

(
RA,ix (0 ≤ x ≤ xi)
RB,i(l − x) (xi ≤ x ≤ l)

(5.11)

ただし，xi[m]は荷重座標，RA,i及びRB,i[N]は支点反力を表します．

RA,i =
l − xi
l
Pi (5.12)

RB,i =
xi
l
Pi (5.13)

また，たわみ yi[m]は曲げの基礎式

∂2y

∂x2
= −Mi (x)

EI
(5.14)

に 5.11式で計算される曲げモーメントを代入して積分することにより，荷

重の両側においてそれぞれ次の式で計算できます．

yi(x) =


RA,i · x

6EI

¡
x2 − 2xil + x2i

¢
(0 ≤ x ≤ xi)

RB,i · l−x6EI

¡
x2 − 2xl + x2i

¢
(xi ≤ x ≤ l)

(5.15)

はりに作用するすべての荷重による結果は，重ね合わせの原理により求

めることができます．以上の式により，任意個数（ここでは最大 10個）の

荷重が作用する場合の SFD，BMD及びたわみ線図を描画するプログラム

を次に示します．

100 !---はりのたわみ線図（両端支持はり・集中荷重の場合）
110 set window -0.1, 1.0, -1.1, 1.1

120 draw grid(0.5, 0.5)

130

140 declare numeric i, j, n, l, dx, E, SMA, Z, xj,&

& smax, bmax, dmax

150 dim x(10), p(10), sfd(101), bmd(101), defl(101)

160

170 !-関数の定義（両支点反力）
180 def RA(i)=(l-x(i))/l*p(i)

190 def RB(i)=x(i)/l*p(i)

200

210 !-データ入力
220 input prompt "はりの長さ [m]を入力して下さい":l

230 let dx=l/100.0

240 input prompt "はりの縦弾性係数を入力して下さい [GPa]を入力し
て下さい":E

250 let E=E*10^9

260 input prompt "はりの断面二次モーメント [mm^4]を入力して下さ
い":SMA

270 let SMA=SMA*10.0^(-12)

280 input prompt "はりの断面係数 [mm^3]を入力して下さい":Z

290 let Z=Z*10.0^(-9)

300 do

310 input prompt "集中荷重の数 [-]を入力して下さい":n

320 loop while n>10

330

340 for i=1 to n

350 input prompt "荷重の位置[m]と大きさ[N]を入力して下さい":&

& x(i), p(i)

360 next i

370

380 !-せん断力・曲げモーメント・たわみの計算
390 mat sfd=ZER

400 mat bmd=ZER

410 mat defl=ZER

420 for j=1 to 101

430 let xj=(j-1)*dx

440 for i=1 to n

450 if xj<=x(i) then

460 let sfd(j)=sfd(j)+RA(i)

470 let bmd(j)=bmd(j)+RA(i)*xj

480 let defl(j)=defl(j)-RA(i)*xj/6.0/E/SMA&

& *(xj^2-2.0*x(i)*l+x(i)^2)

490 else

500 let sfd(j)=sfd(j)-RB(i)

510 let bmd(j)=bmd(j)+RB(i)*(l-xj)

520 let defl(j)=defl(j)-RB(i)*(l-xj)/6.0/E/SMA&

& *(xj^2-2.0*xj*l+x(i)^2)

530 end if

540 next i

550 next j

560

570 !-せん断力・曲げモーメント・たわみの最大値の計算
580 let smax=0.0

590 let bmax=0.0

600 let dmax=0.0

610 for i=1 to 101

620 if abs(sfd(i))>smax then let smax=abs(sfd(i))

630 if abs(bmd(i))>bmax then let bmax=abs(bmd(i))

640 if abs(defl(i))>dmax then let dmax=abs(defl(i))

650 next i

660

670 !-SFDの描画
680 set color 2

690 for i=1 to 101

700 plot lines: (i-1)*dx/l, sfd(i)/smax;

710 next i

720

730 !-BMDの描画
740 plot lines

750 set color 3

760 for i=1 to 101

770 plot lines: (i-1)*dx/l, bmd(i)/bmax;

780 next i

790

800 !-たわみの描画
810 plot lines

820 set color 1

830 for i=1 to 101

840 plot lines: (i-1)*dx/l, -defl(i)/dmax;

850 next i

860

図 5.6: 集中荷重の作用

する両端支持はり

図 5.7: SFD・BMD・た

わみ線図

図 5.8: ピストンピンの

受ける力

870 print "最大曲げモーメントは", bmax/1000.0, "[kN・m]"

880 print "最大曲げ応力は", bmax/z/10.0^6, "[MPa]"

890 print "最大たわみは", dmax*1000.0, "[mm]"

900

910 end

表 5.3: 変数リスト

i, j, xj ループ変数 smax SFDの最大値
n 荷重の個数 入力値 bmax BMDの最大値
l はりの長さ 入力値 dmax たわみの最大値
dx はりの分割長さ x 荷重座標配列 入力値
E 縦弾性係数 入力値 p 荷重配列 入力値

SMA 断面二次モーメント 入力値 sfd SFD配列
Z 断面係数 入力値 bmd BMD配列

defl たわみ配列

プログラムの説明 まず，140及び 150行で変数と配列の宣言をしていま

す．表 5.3に変数一覧を示します．170から 190行は支点反力計算式の定

義です．

210-290行ははりの材質・形状データの入力部であり，300-360行で作用

する荷重の個数及び各荷重の位置と大きさを入力しています．

このプログラムでははりを 100分割し，各座標におけるせん断力，曲げ

モーメント，たわみを求め，重ね合わせの原理により SFD，BMD，たわみ

線図の値を求めています．なお，各線図はそれぞれの最大値で正規化して

描画しています．380-550行が各座標におけるせん断力等の計算，570-650

行で各最大値を求め，670-850行で各線図を描画してます．

例題 長さ 1[m]で， 5.2.1節の例題で調べた断面形状を持つ両端支持はり

があります．左端から 30，55，85[cm]の場所にそれぞれ 300，150，400[kN]

の力が作用するときの各線図を描画し，最大曲げ応力を求めてみましょう．

ただし，はりの縦弾性係数は 206[GPa]とします．

描画した各線図を図 5.7に，他の計算結果を下記に示します．

はりの断面二次モーメント [mm^4]を入力して下さい 430833

はりの断面係数 [mm^3]を入力して下さい 17233

集中荷重の数 [-]を入力して下さい 3

荷重の位置 [m]と大きさ [N]を入力して下さい 0.3, 300000

荷重の位置 [m]と大きさ [N]を入力して下さい 0.55, 150000

荷重の位置 [m]と大きさ [N]を入力して下さい 0.85, 400000

最大曲げモーメントは 110.625 [kN・m]

最大曲げ応力は 6419.36981372947 [MPa]

最大たわみは 131.458900604809 [mm]

課題 1 両端支持はりにおいて，等分布荷重が作用する場合の計算プログ

ラムを作成してみて下さい．

課題 2 片持ちはりにおいて，集中荷重・等分布荷重が作用する場合の計

算プログラムを作成してみて下さい．

課題3 5.1節で考えたピストンクランク機構においては，ピストンピンに

は図 5.8に示すような荷重が作用すると近似できます．コンロッドの小端部

において，力（ここでは慣性力とするが，通常は燃焼圧力による力の方が

大きい）を等分布荷重として支えています．ピストンピンは長さ 40[mm]，

外径 20[mm]，内径 14[mm]の円管とし，コンロッド小端部幅を 20[mm]と

するとき，各線図や最大曲げ応力について計算を行ってみて下さい．

5.3 ガスサイクル

自動車や飛行機などの機械を動かすために，我々は熱機関と呼ばれる機

械を利用しています．熱機関は熱エネルギーを仕事に変換しますが，その

中でも動作流体が気体のものをガスサイクルと呼びます．身近なガソリン

エンジンやディーゼルエンジンなどの往復作動式エンジンでは，燃料を燃

焼させることにより熱を発生，生じた高温高圧ガスによってピストンクラ

ンク機構を作動させ，回転仕事を取り出しています．ここでは，このよう

な往復作動式エンジンのガスサイクルについて調査してみることにします．

5.3.1 理論熱効率

まず，身近な二種類のエンジンについて，熱効率がどのような傾向を持

つのか調査してみましょう．ただし，ここで求めるのは理論熱効率であり，

実機関における損失は考慮していません．

オットーサイクル オットーサイクルは定容サイクルとも呼ばれ，定容的

に加熱が行われるサイクルです．ガソリンエンジン等の火花点火エンジン

においては，吸入した混合気を圧縮し，ちょうど上死点付近で点火栓により

着火，急激な燃焼によりほぼ定容的に熱が加えられると仮定できます．そ

のため，オットーサイクルは火花点火エンジンを近似するサイクルとされ

ています．

ディーゼルサイクル 一方，ディーゼルサイクルは定圧サイクルとも呼ば

れ，定圧的に加熱が行われるサイクルです．ディーゼルエンジン（圧縮着

火エンジン）では空気のみを圧縮し，その高温空気の中へ燃料を噴射しま

す．燃料は空気と混合し，自着火に至ったところから燃えはじめます．燃

焼は緩慢なため，特に低速ディーゼルエンジンでは膨張過程において，ほ

ぼ圧力一定の状態で熱が発生すると仮定できます．（実際にはかなり異なり

ます．）そのため，このサイクルは圧縮着火エンジンを近似するサイクルと

されています．

いずれのエンジンにおいても，排気行程においては下死点で瞬時にガス

の交換が行われ，定容的に排熱されると近似します．また，圧縮・膨張の

各行程は断熱的に行われると仮定します．

オットーサイクル及びディーゼルサイクルの理論熱効率 ηoto，ηdiは，そ

れぞれ次の式により求められます [7, 6]．

ηoto = 1 − 1

²κ−1
(5.16)

ηdi = 1 − 1

²κ−1
σκ − 1
κ (σ − 1) (5.17)

ここで，κは比熱比，²は圧縮比，σは締切比を表します．これらの式を用

い，圧縮比に対する熱効率の変化を計算してみましょう．十進 BASICに

よるプログラムを次に示します．

100 !---理論熱効率の計算
110 set window -2.0, 30.0, -5.0, 100.0

120 draw grid(5.0, 10.0)

130

140 declare numeric eps, kap, zet

150 input prompt "比熱比と締切比を入力して下さい（kap, zet）":kap, zet

160

170 !-オットーサイクルの理論熱効率

図 5.9: オットーサイクル・ディーゼルサイクルにおける理論熱効率の比較

180 plot lines

190 set line color 2

200 for eps=3 to 25

210 plot lines: eps, (1-(1/eps)^(kap-1))*100.0;

220 next eps

230

240 !-ディーゼルサイクルの理論熱効率
250 plot lines

260 set line color 4

270 for eps=3 to 25

280 plot lines: eps,

290 (1-(zet^kap-1)*(1/eps)^(kap-1)/kap/(zet-1))*100.0;

300 next eps

310

320 end

プログラムの説明 もうこの程度のプログラムでは説明は必要ありません

ね．140行は変数の宣言です．リストは表 5.4に示してあります．入力され

た比熱比，締切比において，圧縮比が 3から 25の範囲で各熱効率の計算，

描画を行っています．青線はオットーサイクル，赤線はディーゼルサイク

ルを表します．

表 5.4: 変数リスト
kap 比熱比 入力値 zet 締切比 入力値
eta 圧縮比 入力値

例題 図 5.9に，比熱比が 1.4，締切比が 2の場合における計算結果を示

します．損失を考慮していないため，実際のエンジンではいずれももっと

低い値になりますが，図から，同じ圧縮比で比較した場合には，全体的に

オットーサイクルの方が熱効率が高いことが分かります．

一般的に，ディーゼルエンジンはガソリンエンジンより高い熱効率を持

つ，という話は聞いたことがあると思います．この違いはどういうことで

しょうか？それは，エンジンの運転条件にあります．自動車に用いられて

いるようなガソリンエンジンでは，ノッキング（異常燃焼）を防ぐために

圧縮比が 9から 12程度の範囲に制限されています．一方，ディーゼルエン

ジンにはその制限がないため，通常 16から 21程度の範囲で運転されてい

ます．したがって，このような運転条件を考慮すると，ディーゼルエンジ

ンの方が高い熱効率が得られることがわかります．

課題 1 比熱比を変化させて，その影響について考察してみて下さい．な

ぜ希薄燃焼方式のエンジンは高い熱効率を達成することができるのでしょ

うか？

次節では，これらガスサイクルについてもう少し詳しく調べてみること

にします．本節では最終的な理論熱効率式の比較を行いましたが，サイク

ル中の各状態変化を計算，描画し，それぞれの特徴について詳細な比較を

行ってみましょう．

5.3.2 ガスサイクルの p-v線図と t-s線図

系の状態が変化し，再びもとの状態に戻るとき，そのたどる過程をサイ

クルと呼びます．熱機関のサイクルは通常，いくつかの状態変化によって

構成されています．基準となる数個の状態点を経由し，最初の点まで戻っ

てくる各過程を計算して，仕事や熱量について検討を行ってみましょう．

ガスサイクルの評価を行う際，p-v線図や t-s線図が用いられます．p-v

線図では比体積 v[m3/kg]に対して圧力 p[Pa]をプロットしてあり，サイク

ル線図の囲む面積は仕事 w[J/kg]を表します．また，t-s線図においては

比エントロピ s[J/kg]に対して温度 T [K]をプロット，その囲む面積は熱量

q[J/kg]を表します．

サイクル計算を行うために，まずは一つの状態変化について必要な式を

考えてみましょう．計算にはガス定数や比熱等の物性値が必要です．また，

指定された状態の状態量を明らかにする必要があります．二点間の状態変

化はポリトロープ変化にしたがうとし，変化後の状態量を計算します [7, 6]．

理想気体のガス定数R[J/kg·K]と分子量M [kg/kmol]，一般ガス定数　

R0=8314[J/kmol·K]の間には次の関係があります．

R =
R0
M

(5.18)

また，比熱比 κ[-]が与えられた場合には，ガス定数 Rを用いて定圧比熱

cp[J/kg·K]，定容比熱 cv[J/kg·K]を次の式で計算することができます．

cp =
κR

κ− 1 (5.19)

cv =
R

κ− 1 (5.20)

示量性状態量としては，質量m[kg]，初期の体積 V [m3]のいずれかを与

えます．理想気体の場合，次の状態方程式が成り立ちますから，p，v，T

のうち二つの示強性状態量が指定されると，残りの一つの状態量が決定さ

れます．ただし，未だ与えられていない量を与えることが必要です（例え

ば V が与えられているときには pと T を与え，mを求める）．

pv = RT (5.21)

pV = mRT

また，比内部エネルギ u[J/kg]，比エンタルピ h[J/kg]，比エントロピ

s[J/kg·K]は，基準状態として圧力 P0=0.1013[MPa]，T0=273.15[K]をと

ると，それぞれ次の式で求められます．

u = cv (T − T0) (5.22)

h = cp (T − T0) (5.23)

s = cp ln

µ
T

T0

¶
−R ln

µ
P

P0

¶
(5.24)

さて，状態変化はポリトロープ変化として一般化できます．ポリトロー

プ変化は次式により表され，

pvn = const (5.25)

ポリトロープ指数 nが 0の場合には等圧変化，1の場合には等温変化，比

熱比 κに等しい場合には断熱変化，無限大の場合には等容変化に対応しま

す．また，(5.25)式と理想気体の状態方程式 (5.21)より，次の式が導かれ

ます．

Tvn−1 = const (5.26)

T

P
n−1

n

= const (5.27)

これらの式により，与えられたポリトロープ指数による二点間の状態変化

を計算することができます．初期状態点をのぞき，各経由点の状態量の一

つはこれらの式から求められますから，残り一つの状態量を計算者が与え

て行くことになります．

次に，変化にともない外部に対して行われる仕事，及び加えられる熱量

を計算します．ポリトロープ指数 nが 1及び無限大でない場合には，1[kg]

の理想気体が状態 1-2間で行う仕事 w12[J/kg]は次式で求められます．

w12 =

Z 2

1
pdv =

p1v1
n− 1

"
1 −

µ
v1
v2

¶n−1#
=
p1v1
n − 1

"
1 −

µ
p2
p1

¶n−1
n

#

=
1

n− 1 (p1v1 − p2v2) =
R

n− 1 (T1 − T2) (5.28)

熱量 q12[J/kg]は熱力学の第一法則より次式で求められます．

q12 = cv (T2 − T1) + w12 = cv (T2 − T1) + R

n − 1 (T1 − T2)

=

µ
cv − R

n − 1
¶
(T2 − T1) = cvn − κ

n− 1 (T2 − T1) (5.29)

ポリトロープ指数が 1，すなわち等温変化の場合には仕事と熱量は等しく，

次式で求められます．

w12 = q12 = RT ln
p1
p2
= RT ln

V2
V1

(5.30)

ポリトロープ指数が無限大，すなわち等容変化の場合には仕事wはゼロで，

熱量は次式で表されます．

w12 = 0

q12 = cv (T2 − T1) (5.31)

以上から，サイクル全体における正味仕事W [J]及び系に加えられた熱

量Q[J]は，次式で求められることになります．添え字は各段階における状

態変化を表します．

W = m
nX
i=1

wi (5.32)

Q = m
nX
i=1

Q0i (5.33)

(5.34)

ここで，Qは系に加えられた熱量を表すため，Q0iには次の値を用います．

Q0i =

(
Qi Qi > 0

0 Qi ≤ 0

最後に，サイクル熱効率 η[-]は次式で求めることができます．

η =
W

Q
(5.35)

以上の式を用いて，最大 10個の状態点により構成されるガスサイクル

（もちろんサイクルなので，最後は初期状態点に戻ってこないといけませ

んよ）の解析を行うプログラムを次に示します．今度はちょっと大がかり

ですね．でも，入出力部がややこしいだけで，計算はこれまで示した式を

使っているだけです．入出力部はわかりやすくするため，文献 [8]の例を

参考にさせて頂きました．

1000 !---ガスサイクルの p-v線図と t-s線図
1010 !-副プログラムの宣言
1020 declare external sub Property

1030 declare external sub State

1040 declare external sub UHS

1050 declare external sub Output

1060 declare external sub Process

1070 declare external sub WQ

1080

1090 !-p-v線図表示用座標軸
1100 set window -0.1, 1.0, -0.1, 3.0

1110 draw grid(0.1, 1.0)

1120 !-t-s線図表示用座標軸
1130 !set window -0.1, 0.5, 250.0, 1000.0

1140 !draw grid(0.1, 100.0)

1150

1160 declare numeric nin, ns, si

1170 declare numeric p0, t0, R, mw, k, cp, cv

1180 declare numeric m, vv, n, w, q, wt, qt, dp, dv

1190 declare numeric pn, vn, tn, un, hn, sn

1200 dim p(10), v(10), t(10), u(10), h(10), s(10)

1210 let p0=0.1013*10^6

1220 let t0=273.15

1230

1240 !-データ入力
1250 do

1260 input prompt "計算する状態の数を入力して下さい":ns

1270 loop while ns>10

1280

1290 call Property(R, mw, k, cp, cv)

1300

1310 let m=0.0

1320 let vv=0.0

1330 do

1340 input prompt "示量性状態量として [(1)質量・(2)体積]を入
力します":nin

1350

1360 select case nin

1370 case 1

1380 input prompt "質量 [kg]を入力して下さい":m

1390 if m>0.0 then exit do

1400 case 2

1410 input prompt "体積 [m^3]を入力して下さい":vv

1420 if vv>0.0 then exit do

1430 case else

1440 print "正しい番号を入力して下さい"

1450 end select

1460 loop

1470

1480 let si=1

1490 let pn=0.0

1500 let tn=0.0

1510

1520 print si, "の状態点について入力します"

1530 call State(pn, vv, tn)

1540 call State(pn, vv, tn)

1550

1560 if m=0.0 then let m=pn*vv/R/tn

1570 if pn=0.0 then let pn=m*R*tn/vv

1580 if vv=0.0 then let vv=m*R*tn/pn

1590 if tn=0.0 then let tn=pn*vv/m/R

1600 let vn=vv/m

1610

1620 call UHS(R, cp, cv, pn, tn, un, hn, sn, p0, t0)

1630 let p(si)=pn

1640 let v(si)=vn

1650 let t(si)=tn

1660 let u(si)=un

1670 let h(si)=hn

1680 let s(si)=sn

1690

1700 !-初期状態の出力
1710 print "ガス定数は", R/1000.0, "[kJ/kg/K]"

1720 print "比熱比は", k, "[-]"

1730 print "定圧比熱は", cp/1000.0, "[kJ/kg/K]"

1740 print "定容比熱は", cv/1000.0, "[kJ/kg/K]"

1750 print "質量は", m, "[kg]"

1760 call Output(si, m, p, v, t, u, h, s)

1770

1780 !-状態変化の計算
1790 let wt=0.0

1800 let qt=0.0

1810

1820 for si=2 to ns

1830 print si, "の状態点について入力します"

1840 let pn=0.0

1850 let vv=0.0

1860 let tn=0.0

1870

1880 do

1890 input prompt "ポリトロープ指数 [-]を入力して下さい":n

1900

1910 if n=0.0 then

1920 print "等圧変化です"

1930 let pn=p(si-1)

1940 elseif n=1.0 then

1950 print "等温変化です"

1960 let tn=t(si-1)

1970 elseif n=k then

1980 print "断熱変化です"

1990 let sn=s(si-1)

2000 elseif n>10.0 then

2010 print "等容変化です"

2020 let vn=v(si-1)

2030 let vv=vn*m

2040 end if

2050 loop while n<0.0

2060

2070 call State(pn, vv, tn)

2080 call Process(m, p(si-1), v(si-1), t(si-1), pn, vn, tn, vv, n)

2090 call UHS(R, cp, cv, pn, tn, un, hn, sn, p0, t0)

2100 let p(si)=pn

2110 let v(si)=vn

2120 let t(si)=tn

2130 let u(si)=un

2140 let h(si)=hn

2150 let s(si)=sn

2160 call Output(si, m, p, v, t, u, h, s)

2170 call WQ(si, R, k, cp, cv, m, v, t, n, w, q)

2180 print "W = ", w/1000.0, "[kJ]"

2190 print "Q = ", q/1000.0, "[kJ]"

2200 let wt=wt+w

2210 if q>0.0 then let qt=qt+q

2220

2230 set line color 2

2240 plot lines: v(si-1), p(si-1)/10.0^6;

2250 ! plot lines: s(si-1)/1000.0, t(si-1);

2260 if n>10 then

2270 plot lines: v(si), p(si)/10.0^6;

2280 ! let dp=(p(si)-p(si-1))/100.0

2290 ! for pn=p(si-1) to p(si) step dp

2300 ! let tn=t(si-1)*pn/p(si-1)

2310 ! let sn=cp*log(tn/t0)-R*log(pn/p0)

2320 ! plot lines: sn/1000.0, tn;

2330 ! next pn

2340 else

2350 let dv=(v(si)-v(si-1))/100.0

2360 for vn=v(si-1) to v(si) step dv

2370 let pn=p(si-1)*(v(si-1)/vn)^n

2380 ! let tn=t(si-1)*(v(si-1)/vn)^(n-1.0)

2390 ! let sn=cp*log(tn/t0)-R*log(pn/p0)

2400 plot lines: vn, pn/10.0^6;

2410 ! plot lines: sn/1000.0, tn;

2420 next vn

2430 end if

2440 next si

2450

2460 if qt>0.0 then

2470 print "－－－ガスサイクルの計算結果－－－"

2480 print "仕事は", wt/1000.0, "[kJ]"

2490 print "熱量は", qt/1000.0, "[kJ]"

2500 print "熱効率は", wt/qt*100.0, "[%]"

2510 end if

2520

2530 end

2540

2550 !---熱物性値の計算サブルーチン
2560 external sub Property(R, mw, k, cp, cv)

2570 declare numeric nin, R0

2580 let R0=8.3143

2590

2600 do

2610 input prompt "[(1)分子量・(2)ガス定数]を入力します":nin

2620

2630 select case nin

2640 case 1

2650 input prompt "分子量 [g/mol]を入力して下さい":mw

2660 if mw>0.0 then

2670 let R=R0/mw*1000.0

2680 exit do

2690 end if

2700 case 2

2710 input prompt "ガス定数 [kJ/kg/K]を入力して下さい":R

2720 if R>0.0 then

2730 let mw=R0/R

2740 let R=R*1000.0

2750 exit do

2760 end if

2770 case else

2780 print "正しい番号を入力して下さい"

2790 end select

2800 loop

2810

2820 input prompt "比熱比 [-]を入力して下さい":k

2830 let cp=k*R/(k-1.0)

2840 let cv=R/(k-1.0)

2850

2860 end sub

2870

2880 !---状態量入力サブルーチン
2890 external sub State(pn, vv, tn)

2900 declare numeric nin

2910

2920 do

2930 input prompt "[(1)圧力・(2)体積・(3)温度]を入力します":nin

2940

2950 select case nin

2960 case 1

2970 if pn>0.0 then

2980 print "圧力はすでに与えられています"

2990 else

3000 input prompt "圧力 [MPa]を入力して下さい":pn

3010 let pn=pn*10.0^6

3020 if pn>0.0 then exit do

3030 end if

3040 case 2

3050 if vv>0.0 then

3060 print "体積はすでに与えられています"

3070 else

3080 input prompt "体積 [m^3]を入力して下さい":vv

3090 if vv>0.0 then exit do

3100 end if

3110 case 3

3120 if tn>0.0 then

3130 print "温度はすでに与えられています"

3140 else

3150 input prompt "温度 [K]を入力して下さい":tn

3160 if tn>0.0 then exit do

3170 end if

3180 case else

3190 print "正しい番号を入力して下さい"

3200 end select

3210 loop

3220

3230 end sub

3240

3250 !---比内部エネルギ・比エンタルピ・比エントロピの計算サブルー
チン
3260 external sub UHS(R, cp, cv, pn, tn, un, hn, sn, p0, t0)

3270 let un=cv*(tn-t0)

3280 let hn=cp*(tn-t0)

3290 let sn=cp*log(tn/t0)-R*log(pn/p0)

3300

3310 end sub

3320

3330 !---状態変化の計算サブルーチン
3340 external sub Process(m, p1, v1, t1, p2, v2, t2, vv, n)

3350 if n<10 then

3360 if vv>0.0 then

3370 let v2=vv/m

3380 if p2=0.0 then let p2=p1*(v1/v2)^n

3390 if t2=0.0 then let t2=t1*(v1/v2)^(n-1.0)

3400 elseif p2>0.0 then

3410 if vv=0.0 then let v2=v1*(p1/p2)^(1.0/n)

3420 if t2=0.0 then let t2=t1*(p2/p1)^((n-1.0)/n)

3430 else

3440 if vv=0.0 then let v2=v1*(t1/t2)^(1.0/(n-1.0))

3450 if p2=0.0 then let p2=p1*(t2/t1)^(n/(n-1.0))

3460 end if

3470 else

3480 if t2=0.0 then let t2=t1*p2/p1

3490 if p2=0.0 then let p2=p1*t2/t1

3500 end if

3510

3520 end sub

3530

3540 !---仕事と熱量の計算サブルーチン
3550 external sub WQ(si, R, k, cp, cv, m, v(), t(), n, w, q)

3560 if n=1.0 then

3570 let w=m*R*t(si)*log(v(si)/v(si-1))

3580 let q=w

3590 elseif n>=10 then

3600 let w=0.0

3610 let q=m*cv*(t(si)-t(si-1))

3620 else

3630 let w=m*R*(t(si-1)-t(si))/(n-1.0)

3640 let q=m*cv*(t(si)-t(si-1))*(n-k)/(n-1.0)

3650 end if

3660

3670 end sub

3680

3690 !---状態量の出力サブルーチン
3700 external sub Output(si, m, p(), v(), t(), u(), h(), s())

3710 print "状態点", si

3720 print "圧力は", p(si)/10.0^6, "[MPa]"

3730 print "体積は", v(si)*m, "[m^3]"

3740 print "温度は", t(si), "[K]"

3750 print "比内部エネルギは", u(si)/1000.0, "[kJ/kg/K]"

3760 print "比エンタルピは", h(si)/1000.0, "[kJ/kg/K]"

3770 print "比エントロピは", s(si)/1000.0, "[kJ/kg/K]"

3780

3790 end sub

プログラムの説明 まず，1010-1070行は外部副プログラムの宣言です．こ

れらの定義は後述しますが，そのリストを表 5.6に示します．1090-1140行

は座標軸の設定です．描画するグラフに応じて適切なものを選択，他をコ

メントアウトして下さい．1160-1220行は変数の宣言と初期値の設定です．

変数の一覧は表 5.5に示します．備考の入力値は必ず入力する値，選択値

はメニューから必要に応じて入力する値，既定値は定数を表します．

1250-1270行で計算する状態点の個数を入力し（最終点は第一点と同一

ですが，ここでは一つの点として入力することにしましょう），1290 行

で Propertyを呼び出して熱物性値を計算しています．Propertyの定義は

2550-2860行で，分子量かガス定数のいずれかを入力して他方を計算，ま

た，定圧比熱と定容比熱の計算を行っています．

1310-1460行では，示量性状態量として質量もしくは体積を入力してい

ます．1480-1600行では初期状態点における示強性状態量の入力を行って

います．第一点目であることから二つの状態量を入力し，残りの値を算出

しています．状態量の入力は State関数により行っています．Stateの定義

は 2880-3230行ですが，Stateが一度コールされると，あるステップ（状態

点）における未入力状態量の中から，必ず一つが入力されるようになって

います．また，1620行ではUHS関数によって比内部エネルギ，エンタル

ピ，エントロピの計算を行っています．1630-1680行では第一点目（si=1）

の結果として，各状態量の保存配列に現在の値を格納しています．

ここまでが系全体，また初期状態点に関する計算部です．1700-1760行

で熱物性値の出力，またOutput関数を呼び出すことによって状態量を画

面に出力しています．Outputの定義は 3690-3790行です．

さて，ここからはいよいよ状態変化（目指すはサイクル）の計算です．

1790-1800行では総合の仕事，熱量を初期化しています．1820-2440行が，

各状態変化に関するループです．ある状態点 siに関する値を一つ前の状態

点から求めています．新しい状態点の値を pn，vn，tn変数で表し，それら

は各ステップ，ゼロに初期化されています（すなわち未入力の値はゼロ）．

1880-2050行ではこの状態変化に対するポリトロープ指数を入力してい

ます．入力されたポリトロープ指数が特定の値の場合には，定圧変化や等

温変化，断熱変化に対応しますから，その場合にはそれぞれ一定となる状

態量を新しい変数に設定しています．なお，ポリトロープ指数が無限大の

場合は等容変化を表しますが，プログラム中では無限大を表現できないた

め，10より大きい場合には等容変化であるとしています．

新しい状態点においては，一つ前の状態点における値から状態量が一つ

求まります．そのため，まず 2070行であと一つの状態量を指定していま

す（ここで指定される状態量は，特定の状態変化により設定された状態量

とは異なるものでなくてはいけません）．2080行の Process関数が状態変

化を計算する外部副プログラムです．その定義は 3330-3520行目にありま

す．(5.25)，(5.26)，(5.27)式によって，新しい状態点における未決定の示

強性・示量性状態量を求めています．それぞれ各式を見ながら対応させて

みて下さい．

2090行は比内部エネルギ等の計算です．また，2100-2150行では新しい

状態点における各状態量を配列に保存し，2160行で画面に表示しています．

2170行ではこの状態変化における仕事と熱量の変化を求める関数，WQを

呼び出しています．その定義は 3540-3670行にあり，ポリトロープ指数に

応じた式を用いて計算しています．2200-2210行では総合の仕事量に加算

していますが，熱量については系に加えられた値のみを評価しています．

2230-2430行は状態変化の描画部です．p-v線図，t-s線図のいずれを表

示させたいかによって，コメントアウトする行を選択して下さい．新しい

点までの横軸値を 100分割し，それぞれに対する縦軸値をプロットしてい

ます．自分で確かめてみて下さい．最後に熱効率などを出力してプログラ

ムは終了します．

例題 1 まずはサイクルではなく，ひとつの状態変化について調べてみま

しょう．

表 5.5: 変数リスト
nin メニュー番号の入力に用いる一時変数
ns 状態点の数 入力値
si 現在の状態点番号
p0 初期圧力 既定値
t0 初期温度 既定値
R ガス定数 選択値
R0 一般ガス定数 既定値
mw 分子量 選択値
k 比熱比 入力値
cp 定圧比熱
cv 定容比熱
m 質量 選択値
vv 体積 選択値
n ポリトロープ指数 入力値
w 現在の状態変化における仕事量
q 現在の状態変化における熱量
wt サイクル全体における仕事量
qt サイクル全体において加えられた熱量
dq p-v線図描画時の圧力増分
ds t-s線図描画時のエントロピ増分
pn 状態変化後の圧力 選択値
vn 状態変化後の比体積 選択値
tn 状態変化後の温度 選択値
un 状態変化後の比内部エネルギ
hn 状態変化後の比エンタルピ
sn 状態変化後の比エントロピ
p 各状態点の圧力配列
v 各状態点の比体積配列
t 各状態点の温度配列
u 各状態点の比内部エネルギ配列
h 各状態点の比エンタルピ配列
s 各状態点の比エントロピ配列

表 5.6: 外部副プログラムリスト
Property 熱物性値の計算
State 状態量の入力
UHS 非内部エネルギ，エンタルピ，エントロピの計算
Output 状態量の画面への出力
Process 状態変化の計算
WQ 仕事と熱量の計算

第 5.3.1節で述べたように，ディーゼルエンジンでは空気を吸入・圧縮

し，高温高圧となった空気中へ燃料を噴射することにより自着火させてい

ます．今，直列 6気筒で排気量が 7800[cc]のトラック用過給式ディーゼル

エンジンがあります．燃料の着火温度を約 600[K]とするとき，どの程度の

圧縮比 ²が必要になりますか？ただし，圧縮行程においてはポリトロープ

指数が 1.35のポリトロープ変化を行うとします．また，吸入する空気は過

給により圧力 0.2[MPa]，温度 350[K]であり，分子量は 29[g/mol]，比熱比

は 1.4とします．

この場合，初期の示量性状態量として体積（一気筒あたり 1300[cc]），

示強性状態量として圧力と温度が与えられています．計算すべき状態点数

は，初期状態を含めて二点です．この間の状態変化（指数 1.35によるポリ

トロープ変化）を計算し，到達温度を調査することになります．圧縮比は

² = V1/V2として与えられますから，第 2点目の体積を任意に指定しなが

ら，到達温度が着火点に至る圧縮比を検討することになります．

圧縮比を 17とした場合の計算結果を次に示します．

圧力は 9.24177033350013 [MPa]

体積は 7.60000000000001E-5 [m^3]

温度は 945.504195658091 [K]

この場合には十分着火温度に到達していますね．サイクルとして考えた場

合には，圧縮比は熱効率とも関係してきます．いろいろと計算させて検討

してみて下さい．

課題1 ここで，この例題における到達圧力を見てみましょう．9[MPa]程度

の値になっていることが分かります．等圧サイクルとした場合，この値がサ

イクル最高圧力です．このエンジンは，ボア（ピストン直径）D×ストロー
クSが 115[mm]×125[mm]，コンロッド中心間距離が 200[mm]，また最大出
力時のエンジン回転数が 3000[rpm]であるとします．ピストン質量は1.5[kg]

です．ピストンに作用する慣性力と圧力による力（= pmax × πD2/4），ど

ちらが支配的でしょうか？第 5.1節で作成したプログラムを使って検討し

てみましょう．

課題 2 ある状態点から，状態変化が起こったとします．等圧変化，等温変

化，断熱変化，また任意指数のポリトロープ変化である場合について，そ

の変化の過程（変化後の状態量，仕事，熱量）について調査してみて下さ

い．各変化における p-v線図と t-s線図についても比較してみましょう．例

えば 10気圧で室温の空気が体積 10倍まで膨張した場合にはどのように変

化しますか？

例題 2 次はサイクル計算を行ってみましょう．ここでは，直列 4気筒で排

気量が 1800[cc]，ボア×ストロークが 80[mm]×92[mm]，圧縮比が 10，コ
ンロッド中心間距離が 160[mm]の自然吸気ガソリンエンジン（火花点火式

エンジン）について調査します．作動流体は空気として考え，計算を行っ

てみましょう（これは理想的なサイクル計算における仮定です）．

第 5.3.1節で述べたように火花点火式エンジンはオットーサイクルで近

似できますから，その経路にしたがって入力を進めて行くことにします．

1. 1→ 2：断熱圧縮

2. 2→ 3：等容加熱

3. 3→ 4：断熱膨張

4. 4→ 5：等容冷却

サイクルですから状態点 5は 1のことです．プログラムの都合上，5番目

の点として入力します．

第 1点は初期状態です．吸入空気の圧力は 0.101[MPa]，温度は 300[K]と

します．一気筒辺りの体積は 450[cc]ですね．圧縮後の体積 V2は圧縮比か

ら求められますから，2番目の点の指定量として体積を入力します．この

例では 45[cc]です．等容加熱の過程では燃焼により圧力が急速上昇します．

3番目の点の指定量として圧力を入力しますが，ガソリンエンジンの場合，

この燃焼圧力は経験的に次の式で求めることができます．

pmax[Pa] = (7² − 2) g × 104 (5.36)

g は重力加速度です．この例では約 6.7[MPa]となります．より正しくは，

燃料の発熱量を用いて計算することになります．断熱膨張過程では初期体

積まで膨張しますから，4番目の点の指定量として体積を入力し，第 1点

と同じ体積を与えます．最後に5番目の点の指定量として第 1点の圧力を

入力し，サイクルを完結させます．

このサイクルについて計算した結果を次に示します．また，p-v 線図を

図 5.10に，t-s線図を図 5.11に示します．

－－－ガスサイクルの計算結果－－－

仕事は .281888624415026 [kJ]

熱量は .468336904219724 [kJ]

熱効率は 60.1892829446504 [%]

この計算結果では熱効率が 60% 程度にもなっています（第 5.3.1節の結果

と同様です）．実際には最大でもこの半分程度の効率しか得られません．そ

れは，実機関においては空気サイクルが燃料-空気サイクルになったり（比

図 5.10: p-v線図 図 5.11: t-s線図

熱比の変化），燃焼過程における損失，吸排気における損失，機械的な損

失など，種々の損失が現れてくるからです．ですから，あくまでも定性的

な傾向を捉えるものだと考えて下さい．

課題 2 オットーサイクルとディーゼルサイクルにおいて，それぞれ次の

場合についてサイクルの比較・検討を行ってみて下さい．また，締切比や

比熱比の影響について検討して下さい．

• 圧縮比を揃えた場合

• 最高圧力を揃えた場合

余裕のある人はサバテサイクルについても計算・比較をしてみましょう．

課題 3 本節で取り上げた以外にも，ガスサイクルにはブレイトンサイク

ル，スターリングサイクル，ミラーサイクルなど，種々のサイクルが存在

します．これらについて調査してみて下さい．各サイクルはどのような意

図のもとに構成され，どのような特徴を持つのでしょうか？

サイクル計算に慣れた人は，自分なりのサイクルを作ってみて下さい．本

当に実現・製作できるかどうか？ということを頭において考えてみましょ

う．材料などについては調査が必要だと思いますが，熱機関における現在

の限界と新しいアイデアが見えてくると良いですね．

5.4 流れの数値シミュレーション

近年，内燃機関や航空機の機体といった機械製品の設計・開発段階にお

いて，計算機を用いた予測計算の導入が盛んになっています．特に，熱流

体の計算分野をCFD（Computational Fluid Dynamics）と呼び，汎用的な

計算ソフトウェアも増えてきました．化学反応を伴わない流れの計算（例

えばエンジンシリンダ内の空気流動）であれば，十分実用に耐える段階に

来ており，設計・開発期間の短縮と経費削減につながっています．

本節では，このような流れのシュミレーションを対象にします．ただし，

いきなり本格的な計算は大変ですから，Navier-Stokes方程式を簡略化した

一次元 Burgers方程式を考えることにします．

5.4.1 Burgers方程式

一次元のBurgers方程式は次式で表されます．

∂u

∂t
+ u

∂u

∂x
= ν

∂2u

∂x2
(5.37)

ここで，u[m/s]は速度，t[s]は時間，x[m]は一次元空間における座標を表

します．また，ν [m2/s]はNavier-Stokes方程式における動粘性係数に対応

します．Burgers方程式は簡単な形をしていますが，Navier-Stokes方程式

から圧力勾配項を除いた形に対応しており，非線形対流項と粘性項を持つ

発展型の方程式になっています．非線形項と粘性項の存在は衝撃波を発生

させますので，流れの研究においては重要な基礎方程式です．ここではこ

の衝撃波の発達について観察します．そして，流れの計算法が衝撃波の発

達にどのような影響を与えるのかを調べましょう．

時間発展を伴う現象の観察には，アニメーション表示が効果的です．十

進BASICは簡単にグラフィックの描画ができますから，このような場合に

もとても有効なツールだと言えます．

さて，(5.37)式に示すような時間発展型方程式を計算機で解いてゆく場

合には，まず次のような形に変形を行います．

∂u

∂t
= −u∂u

∂x
+ ν

∂2u

∂x2
(5.38)

上式において左辺は時間微分項ですから，時刻ゼロにおける初期条件u (x, 0)

と系の境界条件を与えれば，後に述べる時間積分法により時々刻々の速度

u (x, t)を求めて行くことができます．ただし，そのためには右辺における

空間座標に対する微分項を適切に評価（離散化）しなくてはいけません．次

節ではこれらの計算手法について述べます．

5.4.2 差分法

差分法は最も古くから，そして今も最も広く利用されている空間離散化

手法です．連続した空間をプログラム上で離散的に（有限の点で）計算する

ため，計算領域を適切な個数に分割し，その格子点上において変数値や微

係数を計算して行きます．詳しい説明は文献に譲ることにして [9, 10, 11]，

ここでは次の二つの差分法を使ってその差を検討することにしましょう．

二次精度中心差分 これは関数 u (x)のテイラー展開により導かれる手法

で，一階，二階の微分値はそれぞれ次式で表されます．

∂u

∂x
=
u (x+∆x, t) − u (x−∆x, t)

2∆x
(5.39)

∂2u

∂x2
=
u (x+∆x, t)− 2u (x, t) + u (x−∆x, t)

∆x2
(5.40)

ここで，∆xは格子幅を表します．計算領域幅をX，分割数をN とすると，

∆x = X/N です．また，この公式を用いる場合，誤差は分割数の 2乗に比

例して減少します．詳しくは上述した文献を参考にして下さい．

一次精度上流差分 後で計算結果をもとに考察しますが，二次精度中心差

分を対流項に用いると計算が不安定化します．その対策として，上流差分

（風上差分とも呼ぶ）が知られています．上流差分を対流項に対して適用す

ると，次のように表すことができます．

µ
u
∂u

∂x

¶
=


u(x, t)

u(x,t)−u(x−∆x,t)
∆x u(x, t) > 0

u(x, t)
u(x+∆x,t)−u(x,t)

∆x u(x, t) < 0

(5.41)

上流差分では，ある格子点の変数値（例えば流速や温度）は流れの上流側

からの影響を主に受けると考えます．したがって，格子点における流速の

正負により，上流側における差分をとって微係数を求めています．

また，(5.41)式は次のように整理することができます．µ
u
∂u

∂x

¶
= u (x, t)

u (x+∆x, t) − u (x−∆x, t)
2∆x

− |u (x, t) |u (x+∆x, t) − 2u (x, t) + u (x −∆x, t)
2∆x

(5.42)

右辺第一項は二次精度中心差分を用いた対流項の形であり，第二項は二階

の二次精度中心差分を用いて次の用に表せます．

1

2
|u|∂

2u

∂x2
∆x (5.43)

これを (5.37)式に適用すると次の用に表されますから，

∂u

∂t
+ u

∂u

∂x
=

µ
ν +

1

2
|u|∆x

¶
∂2u

∂x2
(5.44)

(5.42)式の第二項は数値的な粘性項として作用することになります．実は，

この数値粘性項が時間積分を安定化させることになります．

5.4.3 時間積分法

さて，(5.38)式の右辺における空間微分項の計算ができるようになりま

した．次は時間積分を考えましょう．

時間積分法にも多くの種類があります．一つの分類として，陽解法と陰解

法があります．陽解法では現在明らかな変数値から直接未来の値を求めます

が，陰解法では未来の値が必要となり，連立方程式を解く必要があります．

また，高い精度を求めるときにはルンゲ・クッタ法を用いたり，対流項と

粘性項に違う手法を用いたりします（例えば対流項にはAdams-Bashforth

法，拡散項には Crank-Nicholson法）．

詳細は文献 [9, 10, 11]で学んで頂くこととし，本節では最も簡単な一次

精度オイラー法を用いることにします．ただし，この手法は無条件に不安

定ですので，十分小さな時間刻み∆t[s]をとって計算することにします．

一次精度オイラー法 一次精度オイラーの陽解法は次の様に表されます．

∂F

∂t
= f (x, t)

このとき新しい時間ステップの変数値 F (x, t+∆t)は次式で表されます．

F (x, t+∆t) = ∆t · f (x, t)

したがって，Burgers方程式の場合には次式から新しい時間ステップにお

ける速度を求めて行くことになります．

u (x, t+∆t) = ∆t ·
Ã
−u∂u

∂x
+ ν

∂2u

∂x2

!
t

(5.45)

ここで，時間刻み∆tの決定にはクーラン数C が指標となります．クーラ

ン数は次式で定義され，時間刻み幅の間に格子点を超えて情報が伝わらな

い条件を表します．

C = umax
∆t

∆x
(5.46)

陽解法では少なくともこのクーラン数が 1以下でないと計算が不安定化し

ます（CFL条件）．計算対象・条件や必要精度にもよりますが，通常 0.5

以下に設定します．

5.4.4 差分法の種類による比較

以上の式によりBurgers方程式による衝撃波の時間発展過程を観察して

みましょう．

計算領域は−πから πの区間とし，境界条件には周期境界条件を与えま

す．すなわち，計算領域の両側には同じものがミラーリングされて続いて

いると考えます．プログラム上での計算領域は無限にとれませんから，必

ず端っこがあります．今回の場合は，それが x座標で−πから πの区間で

あり，右端と左端が存在するわけです．ここで，例えば右端の点を考える

と，その点における空間微分値の計算にはそのさらに右隣の値が必要にな

りますよね．周期境界条件では全く同じ計算領域が続いていると考えます

ので，右端の点の隣には，仮想的に左端の点があると考えます．これが周

期境界条件です．

初期条件は次のように与えます．

u (x, 0) = − sin x (5.47)

すなわち，時刻ゼロにおいて領域中心の左側では正の，右側では負の速度

を持っています．領域にある流体塊はどのような変化をすると予測されま

すか？

時間刻みはクーラン数を考慮し，0.01[s]に設定します．最大速度が1[m/s]，

分割数N が 100，∆xが 2π/N とすると，クーラン数は約 0.16です．

この条件の下に，中心差分を用いて作成したプログラムを次に示します．

100 !---バーガース方程式
110 !-時間前進：オイラーの陽解法
120 !-対流項：2次精度中心差分・拡散項：2次精度中心差分
130

140 set window -pi, pi, -2.0, 2.0

150 draw grid(1.0, 0.5)

160

170 declare numeric i, dx, dt, x, t, vis

180 dim v1(100), v2(100)

190

200 let dx=2*pi/100.0

210 let dt=0.01

220 let vis=0.01

230

240 !-初期条件
250 set line color 2

260 for i=1 to 100

270 let x=i-1

280 let v1(i)=-sin(-pi+x*dx)

290 plot lines: -pi+x*dx, v1(i);

300 next i

310 plot lines: pi, v1(1)

320

330 !--- 差分法による計算
340 set line color 1

350 for t=0 to 160

360 for i=1 to 100

370 if i=1 then

380 let v2(1)=v1(1)&

& +dt*(&

& -v1(1)*(v1(2)-v1(100))/2.0/dx&

& +vis*(v1(2)-2.0*v1(1)+v1(100))/dx/dx&

& )

390 elseif i=100 then

400 let v2(100)=v1(100)&

& +dt*(&

& -v1(100)*(v1(1)-v1(99))/2.0/dx&

& +vis*(v1(1)-2.0*v1(100)+v1(99))/dx/dx&

& )

410 else

420 let v2(i)=v1(i)&

& +dt*(&

& -v1(i)*(v1(i+1)-v1(i-1))/2.0/dx&

& +vis*(v1(i+1)-2.0*v1(i)+v1(i-1))/dx/dx&

& )

430 end if

440 next i

450

460 mat v1=v2

470

480 if t>0 and mod(t, 10)=0 then

490 for i=1 to 100

500 let x=i-1

510 plot lines: -pi+x*dx, v2(i);

520 next i

530 plot lines: pi, v2(1)

540 end if

550 next t

560

570 end

一方，対流項に上流差分を用いた場合のプログラム変更点を下記に示し

ます．中心差分の場合のプログラムにおいて，オイラー法による計算部を

置き換えて下さい．

100 for i=1 to 100

110 if i=1 then

120 let v2(1)=v1(1)&

& +dt*(&

& -v1(1)*(v1(2)-v1(100))/2.0/dx&

& +(vis+abs(v1(1))*dx/2.0)&

& *(v1(2)-2.0*v1(1)+v1(100))/dx/dx&

& )

130 elseif i=100 then

140 let v2(100)=v1(100)&

& +dt*(&

& -v1(100)*(v1(1)-v1(99))/2.0/dx&

& +(vis+abs(v1(100))*dx/2.0)

& *(v1(1)-2.0*v1(100)+v1(99))/dx/dx&

& )

150 else

160 let v2(i)=v1(i)&

& +dt*(&

& -v1(i)*(v1(i+1)-v1(i-1))/2.0/dx&

表 5.7: 変数リスト
i doループ変数 t 時間
dx 格子幅 vis 動粘性係数
dt 時間刻み幅 v1 現在の速度配列
x 座標変数 v2 新しい時刻の速度配列

& +(vis+abs(v1(i))*dx/2.0)&

& *(v1(i+1)-2.0*v1(i)+v1(i-1))/dx/dx&

& )

170 end if

180 next i

プログラムの説明 170-180行はいつも通りの変数宣言です．リストは表 5.7

に示します．配列 v1と v2には，各分割点における現在の速度と新しいス

テップの速度を格納しています．v1から v2を求めて配列に格納して行く

ということです．v1(1)には左端，x=-πの速度値，v1(100)には右端から

1格子分手前の値が格納されています．右端の値は周期境界条件より左端

の値と同一ですから，差分計算時にはその値を与えます．

200-220行において，格子幅，時間刻み幅，動粘性係数の設定を行ってい

ます．240-310行では初期条件の設定をし，画面に青色で描画しています．

330-450行が時間に関するループです．時刻ゼロから160ステップ目（160×dt
秒）までを計算しています．360-440行がオイラー法の計算部です．境界条

件によって分岐しており，差分式に対して，左端と右端の場合は周期境界条

件を適用しています．460行では配列の更新を行っています．これは，次の

ステップでは現在求めた新しいステップの値をv1とするためです．480-540

行では 10ステップ毎に速度線図を描画しています．

計算結果 中心差分を用いて計算した結果を図 5.12に，上流差分を用いて

計算した結果を図 5.13に示します．図から，時間が進むにしたがって初期

のサインカーブが立ち上がって行き，衝撃波が形成されて行く様子がわか

ります．

一方，中心差分の場合には，157ステップ（π/2[s]）あたりから速度分布

に不安定性が発生していることが分かります．衝撃波の近傍でギザギザが

発生しています．これは，差分精度が不足することにより発生する現象で

す．図 5.14に，フーリエ・スペクトル選点法により計算を行った結果を拡

大して示します．スペクトル法は周期境界条件の場合に適用でき，比較的

少ない格子点数でも高い精度が得られる手法です．図中，黒線と赤線はス

ペクトル法による結果であり，黒線は 2304分割の場合，赤線は 96分割の

場合を表します．青線は二次精度中心差分で，緑線は一次精度上流差分に

よる結果です．

図から，高精度のスペクトル法においては非常になめらかに計算できて

おり，衝撃波近傍での不安定性はまったく見られません．一方，一次精度

上流差分においても不安定性が観察されません．しかしながら，これは前

述した数値粘性により解が抑え込まれた結果であり，衝撃波発生以前では

むしろ誤差が大きくなっています．解像度を下げたスペクトル法において

は，解全体に渡って弱い波が発生しています．しかしながら，この解は高

精度スペクトル法における解を中心としており，平均的な速度誤差は小さ

いものになります．

このように，差分法一つで計算結果（工学的な予測結果）が変わってく

ることが理解してもらえたと思います．中心差分は基本的には上流差分よ

り精度が高いのですが，不安定さが大きく発散する可能性があります．今

回は言及しませんでしたが，スペクトル法は特殊な条件でしか使うことが

できません．

本節の冒頭で述べたように，最近は設計開発段階で流体計算ソフトウェ

アが用いられることが多くなっています．これらの汎用ソフトウェアを用

いれば，半分はブラックボックスで計算を行うことができます．でも，計算

図 5.12: 対流項に中心

差分を用いた場合

図 5.13: 対流項に上流

差分を用いた場合

図 5.14: スペクトル法

との比較

がうまく行かないときなど，自分で適切な計算法，初期条件，境界条件等

を設定してあげる必要があります．このように，適切な計算法を選択する

こと，これも機械技術者が持っておくと良いノウハウではないでしょうか．

課題 1 時間積分にルンゲ・クッタ法を用いたプログラムを作成し，比較

検討を行って下さい．時間刻み幅や格子幅など，どのような影響が見られ

ますか？

課題 2 Burgers方程式に対するプログラムを参考に，次の一次元対流拡

散方程式を解くプログラムを作成して下さい．

∂c

∂t
+ u

∂c

∂x
= D

∂2c

∂x2
(5.48)

対流拡散方程式では物質濃度 c[kg/m3]の変化を追跡して行きます．D[m2/s]

は拡散係数を表します．また，速度 uは Navier-Stokes方程式などと連立

させ，各格子点における値を随時求めて行かねばならないのですが，ここ

では簡単のために定数としましょう．

境界条件は左端が物質の流入（常に c=1），右端が物質の流出（常に c=0）

とします．初期条件は，左端を除き物質濃度ゼロとします．

また，対流拡散方程式におけるクーラン数は次式で表されます．

C = D
∆t

∆x2
(5.49)

適切な時間刻みを選択して下さい．

これらの条件において拡散係数や速度の影響について計算し，考察して

みて下さい．

5.5 おわりに

以上，4つのテーマについて，十進BASICを使った計算・調査を行って

きました．機械工学の基礎科目それぞれから代表的なテーマを取り上げて

説明してきましたが，他にも計算機を使って調べられる課題がたくさんあ

ります．本章で試みたことを参考に挑戦してみてください．

計算機が教えてくれることは限られていますし，実験から見いだせるこ

とが真実です．でも，簡単な計算によって講義で学んだことを視覚的に確

かめたり，調査したりすることは，理解を深めるのにとても役立ちます．ま

た，新しい発見への道しるべになるかもしれません．計算機は一つの道具

です．うまくつきあって，学習や研究に役立てられることを願っています．

最後になりますが，本章を執筆するのに，関連した多くの書籍を参考に

させて頂きました．ここに記し，謝意を表します．

